Chrysler_Thermal_Shutdown

* dsauersanjose@aol.com 8/15/08
* www.idea2ic.com





Item	D	W	L	Rth_K_deg/Watt
Top_Output_Transistor_Si	7	22	42	2
Copper_Heat_Sink_Below_Cu	22	33	63	1
Cross_Die_Series_Si	1	7	42	.92
Cross_Die_Parallel_Si	3.5	1	42	22.5

*

```
.option set srcsteps = 1

IPWR       0      VPWR   40
RPT        VPWR   V1     2
RCu1       V1     VCu2   .5
RCu2       VCu2   0      .5
RS1        V1     V2     .92
RP1        V1     VCU2   22.5
RS2        V2     V3     .92
RP2        V2     VCU2   22.5
RS3        V3     V4     .92
```

```
RP3        V3     VCU2   22.5
RS4        V4     V5     .92
RP4        V4     VCU2   22.5
RS5        V5     V6     .92
RP5        V5     VCU2   22.5
RS6        V6     V7     .92
RP6        V6     VCU2   22.5
RS7        V7     V8     .92
RP7        V7     VCU2   22.5


.tran     10n    .2m     0        10n

.control
run
set      pensize = 2
dump
.endc

.end
```

======Chrysler_Thermal_Shutdown==========

Up until Chysrler, it was thought that the two worst tests on a power amplifier was to short the output pin in all ways possible with or without a heat sink. What Chysler taught was there might be a narrow range of thermal resistance which is actually worst case for this short circuit test.

Current Value	Node Name	
59.0363951495685	vpwr	
19.0363951495685	v1	
10	vcu2	20watts
17.6328503820013	v2	
16.5414043856093	v3	
15.7174291463177	v4	
15.12723323212	v5	
14.7466841878579	v6	
14.5602217859438	v7	
14.5602217859438	v8	

This simulation models the thermal situation as one volt equal one degree C and one Amps equals one watt. The Thermal resistance model to Ohms. All the thermal gradients can be found from a DC simulation. The data above shows 20 watts being applied to a 22X42 mil NPN transistor which has it's copper heat sink "bolted to a side of a submarine". Node vcu2 is taken to be a crude average temperature (10 degreeC) of the copper heat sink. The NPN power transistor is at 60 degrees. Node V1..V8 represent temperatures take 1 to 8 mils away from the output NPN.

```
   Current Value  Node Name
   -------------  ---------
118.072790299137  vpwr
 38.072790299137  v1
              20  vcu2   40watts
35.2657007640026  v2
33.0828087712185  v3
31.4348582926354  v4
30.2544664642401  v5
29.4933683757158  v6
29.1204435718875  v7
29.1204435718875  v8
```

It is obvious that a large thermal gradient can exist across the die. But with a good heat sink, the temperature at the output will remain below 125C with its power being defined by current limit. As long as the output transistor stays below 125C, it will survive 1000 hours.

With a very poor heat sink, there cannot be much of a thermal gradient across the die, and if the thermal shut down is set under 125C, the die will survive 1000 hours.

Chysrler manage to design their car radio chassis with just the right amount of thermal resistance to destroy an output NPN with a 1000 hour short circuit test. If their chassis had either been a little better or worst, the NPN would have survived.

This problem was due in part to the LM383 being die size limited. If and when possible, it is much better to be able to locate the thermal shutdown circuitry as close as possible to the output transistors.

***#1=============WinSpiceVersion=================**

```
Chrysler_Thermal_Shutdown
* dsauersanjose@aol.com  8/15/08
* www.idea2ic.com
*
*     ^
*    /_\      Zenor TC +3.5mV/C
*    _|_      Diode TC -2.2mV/C
*   /   \I1
*   \/‾\/
*   /\_/\                                Rho_Si = .685 Kdeg/Watt_per_cm
*   \___/      R1                        Rho_Cu = .246 Kdeg/Watt_per_cm
*     |_____/\  /\  /\_         ^ VC     Rth + Rho*D/(W*L)
*     |       \/  \/   |     QN1/_\
*     |                |         _|               L
*    _|_/          ____|_____|'           ____________
*   / ^ VZE       |          |`->      W /           /| D
*    /_\          |             |       /  In Mils  / /
*     |           |      R2     |      /___________/ /
*    _|_          |_/\  /\  /\__|      |___________|/
*    ///              \/  \/    |
*                               |
*                              _|_
*                              ///
*
```




```
*
.option set srcsteps = 1

IPWR      0     VPWR  40
RPT       VPWR  V1    2
RCu1      V1    VCu2  .5
RCu2      VCu2  0     .5
RS1       V1    V2    .92
RP1       V1    VCU2  22.5
RS2       V2    V3    .92
RP2       V2    VCU2  22.5
RS3       V3    V4    .92
RP3       V3    VCU2  22.5
RS4       V4    V5    .92
RP4       V4    VCU2  22.5
RS5       V5    V6    .92
RP5       V5    VCU2  22.5
RS6       V6    V7    .92
RP6       V6    VCU2  22.5
RS7       V7    V8    .92
RP7       V7    VCU2  22.5


.op

.control
run
set     pensize = 2
print   vpwr v1 v2 v3 v3 v5 v6 v7 vcu2
.endc



.end
```